
Calculations in k[ ,..., ]/I where I is an ideal: finite dimensional case.

interface(imaginaryunit=_i):

with(Groebner):

I := [x*y^3-x^2,x^3*y^2-y];

G := Basis(I,tdeg(x,y));

LTI := map(LeadingMonomial,G,tdeg(x,y));

LTC := [1,x,x^2,x^3,y,x*y,x^2*y,x^3*y,y^2,y^2*x,y^2*x^2,y^3];

nops({op(LTC)});

f := 2*x+3*y^2-2;

finv := add( c[i]*LTC[i], i=1..nops(LTC) );

zero := NormalForm( f*finv, G, tdeg(x,y) ) - 1;

The main point here is that if   has an inverse  then   = 1 mod  .  Since  is finite 
dimensional and we know the complement of  <LT(I)>  then we know the form of elements
in  so we can set up a linear system over , which in this example is Q, to solve for the 
coefficients of  .
eqns := {coeffs(zero,[x,y])};

sols := solve(eqns);

Now check it by multiplication in the quotient ring.
NormalForm( f*subs(sols,finv), G, tdeg(x,y) );

G2 := factor( Basis(I,plex(x,y)) );

Clearly y is a zero divisor in the quotient ring so it will not have an inverse.
f := y;

zero := NormalForm( f*finv, G, tdeg(x,y) ) - 1:

eqns := {coeffs(zero,[x,y])};

sols := solve(eqns);

How many solutions are in V where V = V(I) ?
Gy := Groebner[Basis](I,plex(x,y));

fy := Gy[1];

Gx := Groebner[Basis](I,plex(y,x));

fx := Gx[1];

It follows that the variety of I is finite, and there are at most 12 solutions for x and 11 for y
hence |V| <= 132.
These univariate polynomials fx and fy which are in the ideal I can be found directly using 
the following command.  One can compute them using the FGLM algorithm instead of using
the lex monomial ordering to eliminate all variables except 1.
Groebner[UnivariatePolynomial](x,I);

Groebner[UnivariatePolynomial](y,I);

I'll count the number of solutions by computing a prime decomposition of the radical of I.
with(PolynomialIdeals):

I := <I>;

IsRadical(I);

The use of simplify below is to compute Groebner bases for each prime component of J.
map(Simplify,[PrimeDecomposition(Radical(I))]);

From here we see that there are 1 + 4 + 4 + 1 + 1 = 11 solutions. 


